# Bending of a Loading Bridge

## Introduction

This worksheet demonstrates how to use Maple to determine the bending of the loading bridge where a vehicle is handling a mass *m*.

```
> restart;
```

$$Typesetting:-mparsed(m, m) \quad (1.1)$$

## Description of the Loading Bridge Problem

The length in meters of the bridge is L and the distance between the outer wheels of the vehicle is l. The position of the middle of the vehicle from the left side is x0 and its mass in kg is mv. E is the modulus of elasticity and I is the geometical moment of inertia of the loading bridge; the unit of E*I is Nm^2.

The values of the parameters are:

```
> L:=10:
> val:= l=1, x0=8, mv=1000, m=4000, EI=2*10^7:
```

## Problem Solution

The load is modeled as a piecewise function.

```
> load:=piecewise(x<x0-l/2,0,x<x0+l/2,m+mv,0):
```

The differential equation describing the bending of the bridge is:

$$> deq := EI\left(\frac{d^4}{dx^4}\,y(x)\right) = -load$$

$$deq := EI\left(\frac{d^4}{dx^4}\,y(x)\right) = -\left(\begin{cases} 0 & x < x0 - \frac{l}{2} \\ m + mv & x < x0 + \frac{l}{2} \\ 0 & otherwise \end{cases}\right) \quad (3.1)$$

The bridge can be considered as a rod lying on two bearings. There are no bending moments at the ends and the boundary conditions for the bridge are:

```
> boundery:= y(0)=0, y(L)=0, D(D(y))(0)=0, D(D(y))(L)=0:
```

We insert the values of the parameters, solve the differential equation and assign the solution to *y*:

```
> sol:=dsolve(subs(val,{deq,boundery}),y(x)):
> assign(sol):
> y:=unapply(y(x),x);
```

(3.2)

$$y := x \mapsto \begin{cases} \frac{1}{120000} x^3 - \frac{383}{480000} x & x < \frac{15}{2} \\ -\frac{135}{4096} + \frac{77}{240000} x^3 + \frac{16109}{960000} x - \frac{9}{2560} x^2 - \frac{1}{96000} x^4 & x < \frac{17}{2} \\ \frac{257}{12000} - \frac{1}{30000} x^3 - \frac{1057}{120000} x + \frac{1}{1000} x^2 & \frac{17}{2} \leq x \end{cases} \quad (3.2)$$

This gives a function describing the bending line of the loading bridge. To determine the maximal bending we calculate the derivative and search the zeros. The position of the maximal bending is given by:

```
> tmp:=solve(diff(y(x),x)=0,x);
```

$$tmp := \frac{\sqrt{1149}}{6}, -\frac{\sqrt{1149}}{6}, 10 + \frac{\sqrt{429}}{6} \quad (3.3)$$

We transform the zeros to real numbers:

```
> map(evalf,[tmp]);
```

$$[5.649483753, -5.649483753, 13.45205253] \quad (3.4)$$

The only physically possible solution is $\frac{1\sqrt{1149}}{6}$ since the other zeros do not correspond to the bridge. The maximal bending of the bridge (in meters) is given by:

```
> evalf(y(1/6*sqrt(1149)));
```

$$-0.003005211496 \quad (3.5)$$

## Graphical Display of the Solution

Graphical display of the bending of the bridge.

```
> plot(y(x),x=0..L,axes=boxed,
       title="Bending of the Bridge",
       labels=["Distance [m]","Displacement [m]"],
       labeldirections=[default,vertical]);
```




## Conclusion

With the help of Maple we can calculate the exact bending line of the loading bridge. The maximal bending can be calculated for a given set of parameters. This bending line can be used in the calculation of the oscillations of the loading bridge when the load is released suddenly.

## Reference

K. Magnus and H.H. Müller, Grunglagen der Technischen Mechanik, Teubner.